```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the third Weyl algebra $A = A_3(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d1,x1], diff=[d2,x2], diff=[d3,x3],
> polynom=[x1,x2,x3]):
```

and the left A-module M finitely presented by the matrix R defining the divergence operator in 3-space, namely:

```
> R := evalm([[d1,d2,d3]]);
```

$$R := \left[\begin{array}{ccc} d1 & d2 & d3 \end{array}\right]$$

Moreover, let us consider d = d1

```
> d := d1;
```

$$d := d1$$

and the element m of M represented by

```
> lambda := evalm([[0,0,-x1]]);
```

$$\lambda := \left[\begin{array}{ccc} 0 & 0 & -x1 \end{array}\right]$$

Let Q be a matrix with three rows and k columns and entries in A such that ker_A(R.) = Q A^k.

```
> Q := Involution(SyzygyModule(Involution(R,A),A),A);
```

$$Q := \left[\begin{array}{ccc} d3 & d2 & 0 \\ 0 & -d1 & d3 \\ -d1 & 0 & -d2 \end{array}\right]$$

Thus, we have that hom_A(M, A) is isomorphic to ker_A(R.) = Q A^3.

Let us check that (d, m) is a unimodular element of the direct sum of A and M, i.e., d e + λ Q ξ = 1 for certain elements e of A and ξ of A^3, or equivalently that S = (d λ Q) admits a right inverse (e ξ^T)^T. We first have

```
> L := Mult(lambda,Q,A);
```

$$L := \left[\begin{array}{ccc} x1\ d1 & 0 & x1\ d2 \end{array}\right]$$

which yields that S = (d λ Q) is defined by

```
> S := augment(evalm([[d]]),L);
```

$$S := \left[\begin{array}{cccc} d1 & x1\ d1 & 0 & x1\ d2 \end{array}\right]$$

The matrix S admits a right inverse T defined by

```
> T := RightInverse(S,A);
```

$$T := \begin{bmatrix} x1 \\ -1 \\ 0 \\ 0 \end{bmatrix}$$

Hence, we get that e is defined by

```
> e := T[1,1];
```

$$e := x1$$

and ξ is defined by:

```
> xi := submatrix(T,2..4,1..1);
```

$$\xi := \begin{bmatrix} -1 \\ 0 \\ 0 \end{bmatrix}$$

Let us check that μ = Q ξ, namely,

```
> mu := Mult(Q,xi,A);
```

$$\mu := \begin{bmatrix} -d3 \\ 0 \\ d1 \end{bmatrix}$$

is an element of ker_A(R.), i.e., μ induces a left A-homomorphism ϕ from M to A:

```
> Mult(R,mu,A);
```

$$\begin{bmatrix} 0 \end{bmatrix}$$

Now, since $rank_A$(M) = 3 - 1 = 2, let us compute a left A-homomorphism ψ from A to M which is such that ψ(d) + m is a unimodular element of M.

```
> U := ReductionOfUnimodularElement(R,d,lambda,e,mu,A,"splithom"):
```

The output of the command *ReductionOfUnimodularElement* contains two entries

```
> nops(U);
```

$$2$$

the first one U[1], namely,

```
> U[1];
```

$$\begin{bmatrix} 0 & x1+1 & 0 \end{bmatrix}$$

represents an element m_1 of M which is such that ψ(a) = a m_1 for every element a of A. Moreover, the second entry U[2] of U, namely,

```
> U[2];
```

$$\begin{bmatrix} -d3 + d3\, d1^2\, x1^2 + d3\, d1^2\, x1 - x1\, d2 + d2\, x1^2\, d1 + 2\, d3\, d1\, x1 + 2\, d1\, d3 \\ -x1\, d1 - x1^2\, d1^2 + 1 \\ -d1 - d1^3\, x1 - x1^2\, d1^3 - 4\, x1\, d1^2 - 3\, d1^2 \end{bmatrix}$$

induces a left A-homomorphism ϕ from M to A satisfying:

ϕ

$$(\psi(\mathrm{d}) + \mathrm{m}) = (\lambda + \mathrm{d}\ \mathrm{U}[1])\ \mathrm{U}[2] = 1.$$

Finally, let us check again this result. We first have u = λ + d U[1]

```
>  u := simplify(evalm(lambda+Mult(d,U[1],A)));
```

$$u := \left[\begin{array}{ccc} 0 & \mathit{d1} + \mathit{x1}\ \mathit{d1} + 1 & -\mathit{x1} \end{array}\right]$$

then we have u U[2] = 1 since

```
>  Mult(u,U[2],A);
```

$$\left[\begin{array}{c} 1 \end{array}\right]$$

and R U[2] = 0

```
>  Mult(R,U[2],A);
```

$$\left[\begin{array}{c} 0 \end{array}\right]$$

which shows that U[2] defines a left A-homomorphism ϕ from M to A, which satisfies $\phi(\psi(\mathrm{d}) + \mathrm{m}) = 1$.